

Hoare logic and Model checking

Part II: Model checking

Lecture 10: Implementing model checking

Christopher Pulte cp526
University of Cambridge

CST Part II – 2023/24

In the last two lectures we saw LTL and CTL as examples of temporal logics that can specify the behaviour of temporal models. For using temporal logics in the verification of artefacts, we need model checkers to check a temporal model against a temporal logic specification.

In this lecture we will implement a naïve model checker for CTL: the world's worst model checker for CTL.

1

Model checking

What is model checking?

The model checking problem for CTL is to determine, for a given a temporal model M over some set of atomic propositions AP and CTL formula ψ over AP , whether M satisfies ψ :

$$M \models \psi$$

We need a function that computes this.

2

Definite temporal models

The temporal models we have defined in lecture 7 were not restricted in any way that guarantees computability. Here we will assume a **definite temporal model**, using a finite set of states and computable functions for the initial state predicate, the transition relation and the labelling of states.

3

Type of temporal models

```
type state = int

module States = Set.Make(Int)

type 'ap tmodel = {
  s : States.t;           (* finite set *)
  s0 : state -> bool;    (* computable *)
  t : state -> state -> bool; (* computable *)
  l : state -> 'ap -> bool; (* computable *)
}
```

4

Defining a CTL model checker

To check whether the model satisfies a property ψ , we have to check whether the initial states satisfy ψ . We check this using an auxiliary function mca that returns the states satisfying a given state property.

```
let mc (m : 'ap tmodel) (psi : 'ap state_prop) : bool =
  assert (left_total m);
  let v = mca m psi in
  States.for_all (fun s ->
    not (m.s0 s) || States.mem s v
  ) m.s
```

This mca function works by recursion on the formula, calling itself on the sub-formulas.

6

Specifying a CTL model checker

We will implement a naïve CTL model checker:

```
val mc : 'ap tmodel -> 'ap state_prop -> bool
```

which has the following specification:

$$\forall M, \psi. (mc\ M\ \psi \Leftrightarrow M \models \psi)$$

5

CTL model checker

(This is often phrased in terms of “labelling” of states.)

Strategy: For a given CTL state-property ψ : compute the states of the temporal model that satisfies ψ , by

- exploiting CTL formula equivalences to encode ψ as a formula $\hat{\psi}$ that uses only existential path quantification (using negation in the right places)
- (recursively) computing the states satisfying the **sub-formulas** of $\hat{\psi}$, and
- using this information to determine which states should be returned for $\hat{\psi}$.

7

CTL model checker: propositional fragment

mca, for a given temporal model and state property returns the set of states satisfying the state property.

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  match psi with
  | True ->
    m.s
  | False ->
    States.empty
  | AP p ->
    States.filter (fun s -> m.l s p) m.s
  | Not psi' ->
    let v = mca m psi' in
    States.diff m.s v
  ...
```

8

CTL model checker: propositional fragment (continued)

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | And (psi1, psi2) ->
    let v1 = mca m psi1 in
    let v2 = mca m psi2 in
    States.inter v1 v2
  | Or (psi1, psi2) ->
    let v1 = mca m psi1 in
    let v2 = mca m psi2 in
    States.union v1 v2
  | Impl (psi1, psi2) ->
    mca m (Or (Not psi1, psi2))
  ...
```

9

CTL model checker: A

We use

- $A X \psi' = \neg E X (\neg \psi')$
- $A G \psi' = \neg E F (\neg \psi')$

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | A (X psi') ->
    mca m (Not (E (X (Not psi'))))
  | A (G psi') ->
    mca m (Not (E (F (Not psi'))))
  | A (F _) ->
    failwith "TODO:␣exercise"
  | A (U (psi1, psi2)) ->
    failwith "TODO:␣tricky␣exercise"
  ...
```

10

CTL model checker: EX

If we know in which states ψ' holds, then we know in which states $X \psi'$ holds: their predecessors:

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | E (X psi') ->
    let v = mca m psi' in
    States.filter (fun s ->
      States.exists (fun s' ->
        m.t s s'
      ) v
    ) m.s
  ...
```

11

CTL model checker: EF

We use $E F \psi' = E (T U \psi')$

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | E (F psi') ->
    mca m (E (U (True, psi')))
  ...
```

12

Fixpoint. New compared to handout.

```
let rec fixpoint (f : States.t -> States.t)
  (s : States.t) : States.t =
  let s' = f s in
  if States.equal s s' then s else fixpoint f s'
```

14

CTL model checker: EG and EU

Left to do are $E G \psi'$ and $E (\psi_1 U \psi_2)$, which talk about infinite paths. We will implement those using fixpoint operations on sets, where the finite size of the set of states guarantees termination.

13

CTL model checker: EG

For $E G \psi'$:

1. compute the set v of states satisfying ψ'
2. define the output set to be $v' := v$
3. until there are no more changes: remove from v' elements that cannot transition into v'

```
| E (G psi') ->
  let v = mca m psi' in
  fixpoint (fun v' ->
    States.filter (fun s ->
      States.exists (fun s' ->
        m.t s s'
      ) v'
    ) v'
  ) v
  ...
```

15

CTL model checker: EU

For $E(\psi_1 U \psi_2)$:

1. compute the sets v_1 and v_2 of states satisfying ψ_1 and ψ_2
2. define the output set to be $v' := v_2$
3. until there are no more changes: add states from v_1 that can transition into v'

```
| E (U (psi1, psi2)) ->
  let v1 = mca m psi1 in
  let v2 = mca m psi2 in
  fixpoint (fun v' ->
    States.union v'
      (States.filter (fun s ->
        States.exists (fun s' ->
          m.t s s'
        ) v'
      ) v1)
  ) v2
```

16

Counterexamples.

Defs simplified and fixed wrt handout

Actually implementing model checking

This is not very efficient!

In practice,

- the labelling (the v s) are memoised: in our code the v s are re-computed each time, in the case of nested CTL formulas
- “symbolic model checking” uses binary decision diagrams (BDDs) to represent sets of states, and performs operations on sets-as-BDDs, instead of explicitly manipulating the sets;
- the states can be computed lazily;
- “partial order reduction” tries to not enumerate redundant interleavings;
- ...
- 40+ years of tricks!

17

Generating counterexamples

Adapted from “Tree-Like Counterexamples in Model Checking”.

If the specification is not satisfied, and is in ACTL, then we can do better than just say “no”: we can produce a counterexample.

The idea is that $M \not\models \psi^{\text{ACTL}}$ is equivalent to $M \models \neg\psi^{\text{ACTL}}$, where $\neg\psi^{\text{ACTL}}$ can be expressed in ECTL.

So $M \not\models \psi^{\text{ACTL}}$ implies the existence of a witness for the corresponding ECTL property.

We will now assume formulas in negation normal form: formulas without implication, and where the only use of negation is immediately preceding an atomic proposition.

18

Shape of ECTL witnesses

The shape of an ECTL witness for a set of atomic propositions AP and temporal model M :

$$\begin{aligned} \text{Witness}_M &:= \\ &| \text{WAP} \in M.S \rightarrow \text{Witness}_M \\ &| \text{WNAP} \in M.S \rightarrow \text{Witness}_M \\ &| \text{WAnd} \in \text{Witness}_M \rightarrow \text{Witness}_M \rightarrow \text{Witness}_M \\ &| \text{WOrL} \in \text{Witness}_M \rightarrow \text{Witness}_M \\ &| \text{WOrR} \in \text{Witness}_M \rightarrow \text{Witness}_M \\ &| \text{WX} \in M.S \rightarrow M.S \rightarrow \text{Witness}_M \rightarrow \text{Witness}_M \\ &| \text{WF} \in \text{list } M.S \rightarrow \text{Witness}_M \rightarrow \text{Witness}_M \\ &| \text{WG} \in \text{list } (M.S \times \text{Witness}_M) \rightarrow \text{Witness}_M \\ &| \text{WU} \in \text{list } (M.S \times \text{Witness}_M) \rightarrow \text{Witness}_M \end{aligned}$$

There are (on purpose) no cases for A . . .

19

Being an ECTL witness: atomic propositions

A witness for an atomic proposition is just the fact that the atomic proposition holds according to $M.l$:

$$(s \models_M p) \text{ wit-by } W \stackrel{\text{def}}{=} \\ W = \text{WAP } s \wedge M.l \ s \ p$$

Similarly for negation of atomic propositions.

21

Being an ECTL witness

We will define when a witness is a “valid witness” for an ECTL property:

$$(s \models_M \psi) \text{ wit-by } W$$

should hold whenever W is a valid witness for the fact that ψ holds in state s of temporal model M .

20

Being an ECTL witness: next

A witness for ‘next’ is a transition from the current state to a next state, and a witness that the sub-property holds in the next state:

$$(s \models_M (E X \psi)) \text{ wit-by } W \stackrel{\text{def}}{=} \\ \exists s' \in M.S, W' \in \text{Witness}_M. \\ \left(\begin{array}{l} W = \text{WX } s \ s' \ W' \wedge \\ s \ M.T \ s' \wedge \\ (s' \models_M \psi) \text{ wit-by } W' \end{array} \right)$$

22

Being an ECTL witness: future

A witness for the 'future' temporal operator is a finite path that leads to a state for which we have a witness that it satisfies the sub-property:

$$(s \models_M E F \psi) \text{ wit-by } W \stackrel{\text{def}}{=} \begin{array}{l} \exists s' \in M.S, \pi \in \text{list } M.S, W' \in \text{Witness}_M. \\ \left(\begin{array}{l} W = \text{WF } \pi \ W' \wedge \\ \text{IsFinitePath } M \ \pi \wedge \\ \text{nth } \pi \ 0 = s \wedge \\ \text{last } \pi = s' \wedge \\ (s' \models_M \psi) \text{ wit-by } W' \end{array} \right) \end{array}$$

23

Being an ECTL witness: generally

A witness for the 'generally' temporal operator is a lasso-shaped path, together with witnesses that each state along the path satisfies the sub-property:

$$(s \models_M E G \psi) \text{ wit-by } W \stackrel{\text{def}}{=} \begin{array}{l} \exists \text{SWs} \in \text{list } (M.S \times \text{Witness}_M). \\ \left(\begin{array}{l} W = \text{WG } \text{SWs} \wedge \\ \text{let } \pi = \text{firsts } \text{SWs} \text{ in} \\ \text{IsFinitePath } M \ \pi \wedge \\ \text{nth } \pi \ 0 = s \\ (\exists i \in \mathbb{N}. (\text{last } \pi) \ M.T \ (\text{nth } \pi \ i)) \wedge \\ \left(\forall j \in \mathbb{N}, s' \in M.S, W' \in \text{Witness}_M. \right. \\ \left. \left(\begin{array}{l} \text{nth } \text{SWs } j = \langle s', W' \rangle \Rightarrow \\ (s' \models_M \psi) \text{ wit-by } W' \end{array} \right) \right) \end{array} \right) \end{array}$$

24

Being an ECTL witness: until

$$(s \models_M E (\psi_1 U \psi_2)) \text{ wit-by } W \stackrel{\text{def}}{=} \begin{array}{l} \exists \text{SWs} \in \text{list } (M.S \times \text{Witness}_M), s' \in M.S, W' \in \text{Witness}_M. \\ \left(\begin{array}{l} W = \text{WU } (\text{SWs} ++ [\langle s', W' \rangle]) \wedge \\ \text{let } \pi = \text{firsts } \text{SWs} ++ [s'] \text{ in} \\ \text{IsFinitePath } M \ \pi \wedge \\ \text{nth } \pi \ 0 = s \wedge \\ \left(\forall i \in \mathbb{N}, s'' \in M.S, W'' \in \text{Witness}_M. \right. \\ \left. \left(\begin{array}{l} \text{nth } \text{SWs } i = \langle s'', W'' \rangle \Rightarrow \\ (s'' \models_M \psi_1) \text{ wit-by } W'' \end{array} \right) \right) \wedge \\ ((s' \models_M \psi_2) \text{ wit-by } W') \end{array} \right) \end{array}$$

25

Being an ECTL witness: conjunction

$$(s \models_M \psi_1 \wedge \psi_2) \text{ wit-by } W \stackrel{\text{def}}{=} \begin{array}{l} \exists W_1 \in \text{Witness}_M, W_2 \in \text{Witness}_M. \\ \left(\begin{array}{l} W = \text{WAnd } W_1 \ W_2 \wedge \\ (s \models_M \psi_1) \text{ wit-by } W_1 \wedge (s \models_M \psi_2) \text{ wit-by } W_2 \end{array} \right) \end{array}$$

26

Being an ECTL witness: disjunction

$$(s \models_M \psi_1 \vee \psi_2) \text{ wit-by } W \stackrel{\text{def}}{=} \\ \exists W' \in \text{Witness}_M. \\ \left(\begin{array}{l} \left(W = \text{WO}rL \ W' \wedge (s \models_M \psi_1) \text{ wit-by } W' \right) \vee \\ \left(W = \text{WO}rR \ W' \wedge (s \models_M \psi_2) \text{ wit-by } W' \right) \end{array} \right)$$

27

Witnesses beyond ECTL

Can we have witnesses for more than just ECTL?

Yes. For example, one of the nice things about LTL is that counterexamples are just paths.

29

Satisfiability and existence of witnesses

Here we have required finite temporal models, and so witnesses are finite. (Otherwise, we would need to deal with infinite witnesses.)

Now, if we have $M \not\models \psi$ for some ACTL formula ψ , there exists a witness W for the fact that the ECTL formula corresponding to $\neg\psi$ holds — and we could effectively find it by tweaking our model checking algorithm (details elided).

28

Summary

We saw a model checking algorithm for CTL, and sketched how it could be modified to generate counterexamples for ACTL formulas.

30